

ACTIVITE 2

- **sujet** : décomposition de fractions:
- **auteur** : Jean Sainson . Lycée Berthelot Châtellerault.

Une décomposition particulière pour une fraction

Toute fraction peut se décomposer en une somme finie de termes ayant les particularités ci-dessous :

- L'un des termes est un nombre entier relatif éventuellement nul,
- Tous les autres termes sont des fractions de numérateur 1 et de dénominateurs distincts deux à deux.

Pour se convaincre du résultat annoncé, faire exécuter la fonction ci-dessous sur différents exemples. Etudier l'algorithme ci-dessous et s'en inspirer pour démontrer le résultat.

```
> decompose := proc( f :: fraction )
> local x, a, b, q, n, L ;
> option 'Copyright 06/02/2000 - J. Sainson Lycée Marcelin Berthelot 86100 Châtellerault France';
> n := floor( f ) ; # floor est la fonction "partie entière"
> x := f - n ;
> a := numer( x ) ;
> b := denom(x) ;
> L := [ n ] ; # L va contenir la liste des termes de la décomposition
> while a <> 1 do
> q := iquo( b, a ) ; # iquo est la fonction "quotient entier"
> L := [ op(L), 1/(q + 1) ] ;
> x := x - 1/(q + 1) ;
> a := numer( x ) ;
> b := denom( x ) ;
> od ;
> L := [ op(L), 1/b ] ;
> L ;
> end:
> dec1 := decompose( 151 / 213 ) ;
      dec1 := [0,  $\frac{1}{2}$ ,  $\frac{1}{5}$ ,  $\frac{1}{113}$ ,  $\frac{1}{14159}$ ,  $\frac{1}{262148440}$ ,  $\frac{1}{89338345710615240}$ ]
> add( i , i = dec1 ) ; # addition des termes de la liste dec1 pour vérifier
       $\frac{151}{213}$ 
```

```

> dec2 := decompose( 79 / 239 ) ;
      dec2 := [0,  $\frac{1}{4}$ ,  $\frac{1}{13}$ ,  $\frac{1}{277}$ ,  $\frac{1}{93043}$ ,  $\frac{1}{9151592512}$ ,  $\frac{1}{244275632715790612075}$ ,
       $\frac{1}{179011754216099498402549063771393616804800}$ ]
> add( i, i = dec2 ) ;
       $\frac{79}{239}$ 
> dec3 := decompose( -53/11 ) ;
      dec3 := [-5,  $\frac{1}{6}$ ,  $\frac{1}{66}$ ]
> add( i, i = dec3 ) ;
       $\frac{-53}{11}$ 

```

Une version récursive de la fonction ci-dessus.

Remarque: La récursivité de la fonction *decomp* étant "terminale", la version itérative est préférable.

```

> #===== Une fonction locale =====
> decomp := proc( frac :: fraction, L_in :: list )
> local a, b, q, L ;
> option 'Copyright 06/02/2000 - J. Sainson Lycée Marcelin Berthelot 86100 Châtellerault France';
> a := numer( frac ) ;
> b := denom( frac ) ;
> if a = 1 then [ op( L_in ), 1/b ]
> else
> q := iquo( b, a ) ;
> L := [ op( L_in ), 1/( q + 1 ) ] ;
> decomp( frac - 1/( q + 1 ) , L ) ;
> fi ;
> end :
> #===== Fin de la fonction locale =====
> #+++++++ La Fonction Principale ++++++
> decomprec := proc( f :: fraction )
> local x, n ;
> option 'Copyright 06/02/2000 - J. Sainson Lycée Marcelin Berthelot 86100 Châtellerault France';
> n := floor( f ) ;
> x := f - n ;
> decomp( x, [ n ] ) ;
> end:
> #+++++++ Fin de La Fonction Principale ++++++
> dec4 := decomprec( -7777/2943 ) ;
      dec4 := [-3,  $\frac{1}{3}$ ,  $\frac{1}{42}$ ,  $\frac{1}{3170}$ ,  $\frac{1}{16326293}$ ,  $\frac{1}{533095669917405}$ ]

```

```
> add( i, i = dec4 ) ;
```

$$\frac{-7777}{2943}$$